orthog - Orthogonalize variables and compute orthogonal polynomials

Description Syntax Remarks and examples Also see Quick start Options for orthog Methods and formulas Menu Options for orthpoly References

Description

orthog orthogonalizes a set of variables, creating a new set of orthogonal variables (all of type double), using a modified Gram-Schmidt procedure (Golub and Van Loan 2013). The order of the variables determines the orthogonalization; hence, the "most important" variables should be listed first.

Execution time is proportional to the square of the number of variables. With many (>10) variables, orthog will be fairly slow.

orthpoly computes orthogonal polynomials for one variable.

Quick start

Generate ox1, ox2, and ox3 containing orthogonalized versions of x1, x2, and x3 orthog x1 x2 x3, generate(ox1 ox2 ox3)

Same as above

orthog x1 x2 x3, generate(ox*)

Generate op1, op2, and op3 containing degree 1, 2, and 3 orthogonal polynomials for x1 orthpoly x1, generate(op1 op2 op3) degree(3)

Same as above

orthpoly x1, generate(op1-op3) degree(3)

Same as above, and generate matrix op containing coefficients of the orthogonal polynomials orthpoly x1, generate(op1-op3) degree(3) poly(op)

Menu

orthog

Data > Create or change data > Other variable-creation commands > Orthogonalize variables

orthpoly

 $\mathsf{Data}>\mathsf{Create}$ or change data > Other variable-creation commands > Orthogonal polynomials

Title

Syntax

Orthogonalize variables

orthog [varlist] [if] [in] [weight], generate(newvarlist) [matrix(matname)]

Compute orthogonal polynomial

orthpoly varname [if] [in] [weight],

{ generate(*newvarlist*) | poly(*matname*) } [<u>d</u>egree(#)]

orthpoly requires that generate(*newvarlist*) or poly(*matname*), or both, be specified. *varlist* may contain time-series operators; see [U] **11.4.4 Time-series varlists**. iweights, aweights, fweights, and pweights are allowed, see [U] **11.1.6 weight**.

Options for orthog

🛾 Main 🛾

- generate(newvarlist) is required. generate() creates new orthogonal variables of type double. For orthog, newvarlist will contain the orthogonalized varlist. If varlist contains d variables, then so will newvarlist. newvarlist can be specified by giving a list of exactly d new variable names, or it can be abbreviated using the styles newvar1-newvard or newvar*. For these two styles of abbreviation, new variables newvar1, newvar2, ..., newvard are generated.
- matrix (matname) creates a $(d+1) \times (d+1)$ matrix containing the matrix R defined by X = QR, where X is the $N \times (d+1)$ matrix representation of varlist plus a column of ones and Q is the $N \times (d+1)$ matrix representation of newvarlist plus a column of ones (d =number of variables in varlist, and N = number of observations).

Options for orthpoly

Main

- generate(newvarlist) or poly(), or both, must be specified. generate() creates new orthogonal variables of type double. newvarlist will contain orthogonal polynomials of degree 1, 2, ..., d evaluated at varname, where d is as specified by degree(d). newvarlist can be specified by giving a list of exactly d new variable names, or it can be abbreviated using the styles newvar1newvard or newvar*. For these two styles of abbreviation, new variables newvar1, newvar2, ..., newvard are generated.
- poly(matname) creates a $(d + 1) \times (d + 1)$ matrix called matname containing the coefficients of the orthogonal polynomials. The orthogonal polynomial of degree $i \le d$ is

matname[i, d+1] + matname[i, 1]*varname + matname[i, 2]*varname² + ... + matname[i, i]*varnameⁱ

The coefficients corresponding to the constant term are placed in the last column of the matrix. The last row of the matrix is all zeros, except for the last column, which corresponds to the constant term.

degree (#) specifies the highest-degree polynomial to include. Orthogonal polynomials of degree 1, 2, ..., d = # are computed. The default is d = 1.

Remarks and examples

stata.com

Orthogonal variables are useful for two reasons. The first is numerical accuracy for highly collinear variables. Stata's regress and other estimation commands can face much collinearity and still produce accurate results. But, at some point, these commands will drop variables because of collinearity. If you know with certainty that the variables are not perfectly collinear, you may want to retain all their effects in the model. If you use orthog or orthpoly to produce a set of orthogonal variables, all variables will be present in the estimation results.

Users are more likely to find orthogonal variables useful for the second reason: ease of interpreting results. orthog and orthpoly create a set of variables such that the "effects" of all the preceding variables have been removed from each variable. For example, if we issue the command

```
. orthog x1 x2 x3, generate(q1 q2 q3)
```

the effect of the constant is removed from x1 to produce q1; the constant and x1 are removed from x2 to produce q2; and finally the constant, x1, and x2 are removed from x3 to produce q3. Hence,

$$\begin{aligned} \mathbf{q} &1 = r_{01} + r_{11} \, \mathbf{x} \mathbf{1} \\ \mathbf{q} &2 = r_{02} + r_{12} \, \mathbf{x} \mathbf{1} + r_{22} \, \mathbf{x} \mathbf{2} \\ \mathbf{q} &3 = r_{03} + r_{13} \, \mathbf{x} \mathbf{1} + r_{23} \, \mathbf{x} \mathbf{2} + r_{33} \, \mathbf{x} \mathbf{3} \end{aligned}$$

This effect can be generalized and written in matrix notation as

$$X = QR$$

where X is the $N \times (d+1)$ matrix representation of varlist plus a column of ones, and Q is the $N \times (d+1)$ matrix representation of newvarlist plus a column of ones (d =number of variables in varlist and N = number of observations). The $(d+1) \times (d+1)$ matrix R is a permuted upper-triangular matrix, that is, R would be upper triangular if the constant were first, but the constant is last, so the first row/column has been permuted with the last row/column. Because Stata's estimation commands list the constant term last, this allows R, obtained via the matrix() option, to be used to transform estimation results.

Example 1: orthog

Consider Stata's auto.dta dataset. Suppose that we postulate a model in which price depends on the car's length, weight, headroom, and trunk size (trunk). These predictors are collinear, but not extremely so—the correlations are not that close to 1:

```
. use https://www.stata-press.com/data/r18/auto
(1978 automobile data)
. correlate length weight headroom trunk
(obs=74)
                           weight headroom
                 length
                                               trunk
      length
                 1.0000
                           1.0000
      weight
                 0.9460
    headroom
                 0.5163
                           0.4835
                                    1.0000
                 0.7266
                           0.6722
                                    0.6620
                                              1.0000
       trunk
```

regress certainly has no trouble fitting this model:

Source	SS	df	MS	Number o	f obs =	1-1
 				• F(4, 69)	=	10.20
Model	236016580	4	59004145	Prob > F	=	0.0000
Residual	399048816	69	5783316.17	R-square	d =	0.3716
 				• Adj R-sq	uared =	0.3352
Total	635065396	73	8699525.97	Root MSE	=	2404.9
price	Coefficient	C+ J			0.5%	
 price	COETICIENT	Std. err.	t	P> t [95% conf.	interval]
 length	-101.7092	42.12534			95% conf. 185.747	-17.67147
 			-2.41	0.018 -		
 length	-101.7092	42.12534	-2.41 4.24	0.018 - 0.000 2	185.747	-17.67147
 length weight	-101.7092 4.753066	42.12534 1.120054	-2.41 4.24 -1.60	0.018 - 0.000 2 0.114 -1	185.747 .518619	-17.67147 6.987512
 length weight headroom	-101.7092 4.753066 -711.5679	42.12534 1.120054 445.0204	-2.41 4.24 -1.60 1.04	0.018 - 0.000 2 0.114 -1 0.303 -1	185.747 .518619 599.359	-17.67147 6.987512 176.2236

. regress price length weight headroom trunk

However, we may believe a priori that length is the most important predictor, followed by weight, headroom, and trunk. We would like to remove the "effect" of length from all the other predictors, remove weight from headroom and trunk, and remove headroom from trunk. We can do this by running orthog, and then we fit the model again using the orthogonal variables:

- . orthog length weight headroom trunk, gen(olength oweight oheadroom otrunk)
- > matrix(R)
- . regress price olength oweight oheadroom otrunk

888							
Source	SS	df	MS	Number of obs		=	74
				F(4,	69)	=	10.20
Model	236016580	4	59004145	Prob	> F	=	0.0000
Residual	399048816	69	5783316.17	R-sq	uared	=	0.3716
				Adi	R-squared	=	0.3352
Total	635065396	73	8699525.97	0	-	=	2404.9
	, 						
price	Coefficient	Std. err.	t	P> t	[95% co	nf.	interval]
	Coefficient 1265.049	Std. err.		P> t	[95% co 707.345		interval] 1822.753
price olength oweight			4.53			4	
olength	1265.049	279.5584	4.53 4.21	0.000	707.345	4 7	1822.753
olength oweight	1265.049 1175.765	279.5584 279.5584	4.53 4.21 -1.25	0.000	707.345	4 7 5	1822.753 1733.469
olength oweight oheadroom	1265.049 1175.765 -349.9916	279.5584 279.5584 279.5584	4.53 4.21 -1.25 1.04	0.000 0.000 0.215	707.345 618.061 -907.695	4 7 5 2	1822.753 1733.469 207.7122

Using the matrix R, we can transform the results obtained using the orthogonal predictors back to the metric of original predictors:

Technical note

The matrix R obtained using the matrix() option with orthog can also be used to recover X (the original *varlist*) from Q (the orthogonalized *newvarlist*), one variable at a time. Continuing with the previous example, we illustrate how to recover the trunk variable:

4

- . matrix C = R[1..., "trunk"]
- . matrix score double rtrunk = C
- . compare rtrunk trunk

	Count	Minimum	- Difference · Average	Maximum
rtrunk>trunk	74	8.88e-15	1.91e-14	3.55e-14
Jointly defined	74	8.88e-15	1.91e-14	3.55e-14
Total	74			

Here the recovered variable rtrunk is almost exactly the same as the original trunk variable. When you are orthogonalizing many variables, this procedure can be performed to check the numerical soundness of the orthogonalization. Because of the ordering of the orthogonalization procedure, the last variable and the variables near the end of the *varlist* are the most important ones to check.

The orthpoly command effectively does for polynomial terms what the orthog command does for an arbitrary set of variables.

Example 2: orthpoly

Again consider the auto.dta dataset. Suppose that we wish to fit the model

$$mpg = \beta_0 + \beta_1 weight + \beta_2 weight^2 + \beta_3 weight^3 + \beta_4 weight^4 + \epsilon$$

We will first compute the regression with natural polynomials:

. generate double w1 = weight . generate double w2 = w1*w1 . generate double w3 = w2*w1 . generate double w4 = w3*w1 correlate w1-w4 (obs=74) ωЗ ω4 w1 w2 1.0000 w1 0.9915 1.0000 w2 0.9665 0.9916 1.0000 ωЗ ω4 0.9279 0.9679 0.9922 1.0000 regress mpg w1-w4 Source SS df MS Number of obs 74 F(4, 69) = 36.06 Model 1652.73666 4 413.184164 Prob > F = 0.0000 Residual 790.722803 69 11.4597508 R-squared = 0.6764 Adj R-squared = 0.6576 Total 2443.45946 73 33.4720474 Root MSE = 3.3852 mpg Coefficient Std. err. t P>|t| [95% conf. interval] พ1 .0289302 .1161939 0.25 0.804 -.2028704.2607307 ω2 -.0000229 .0000566 -0.400.687 -.0001359 .0000901 wЗ 5.74e-09 1.19e-08 0.48 0.631 -1.80e-08 2.95e-08 w4 -4.86e-13 9.14e-13 -0.53 0.596 -2.31e-12 1.34e-12 23.94421 86.60667 0.28 0.783 -148.8314196.7198 _cons

Some of the correlations among the powers of weight are very large, but this does not create any problems for regress. However, we may wish to look at the quadratic trend with the constant removed, the cubic trend with the quadratic and constant removed, etc. orthpoly will generate polynomial terms with this property:

```
. orthpoly weight, generate(pw*) deg(4) poly(P)
```

. regress mpg pw1-pw4

Source	SS	df	MS	Number of obs F(4, 69)	=	74 36.06
Model Residual	1652.73666 790.722803	4 69	413.184164 11.4597508	Prob > F R-squared	=	
Total	2443.45946	73	33.4720474	Adj R-squared Root MSE	=	
mpg	Coefficient	Std. err.	t	P> t [95% co	onf.	interval]
pw1 pw2 pw3 pw4 _cons	-4.638252 .8263545 3068616 209457 21.2973	.3935245 .3935245 .3935245 .3935245 .3935245 .3935245	2.10 -0.78 -0.53	0.000 -5.4233 0.039 .04129 0.438 -1.0919 0.596994516 0.000 20.512	47 21 58	-3.853192 1.611414 .4781982 .5756028 22.08236

Compare the *p*-values of the terms in the natural polynomial regression with those in the orthogonal polynomial regression. With orthogonal polynomials, it is easy to see that the pure cubic and quartic trends are not significant and that the constant, linear, and quadratic terms each have p < 0.05.

The matrix P obtained with the poly() option can be used to transform coefficients for orthogonal polynomials to coefficients for natural polynomials:

Methods and formulas

orthog's orthogonalization can be written in matrix notation as

$$X = QR$$

where X is the $N \times (d+1)$ matrix representation of *varlist* plus a column of ones and Q is the $N \times (d+1)$ matrix representation of *newvarlist* plus a column of ones (d =number of variables in *varlist*, and N = number of observations). The $(d+1) \times (d+1)$ matrix R is a permuted upper-triangular matrix; that is, R would be upper triangular if the constant were first, but the constant is last, so the first row/column has been permuted with the last row/column.

Q and R are obtained using a modified Gram-Schmidt procedure; see Golub and Van Loan (2013, 254–255) for details. The traditional Gram-Schmidt procedure is notoriously unsound, but the modified procedure is good. orthog performs two passes of this procedure.

orthpoly uses the Christoffel-Darboux recurrence formula (Abramowitz and Stegun 1964).

4

Both orthog and orthpoly normalize the orthogonal variables such that

$$Q'WQ = MI$$

where $W = \text{diag}(w_1, w_2, \dots, w_N)$ with weights w_1, w_2, \dots, w_N (all 1 if weights are not specified), and M is the sum of the weights (the number of observations if weights are not specified).

References

Abramowitz, M., and I. A. Stegun, ed. 1964. Handbook of Mathematical Functions with Formulas, Graphs, and Mathematical Tables. Washington, DC: National Bureau of Standards.

Golub, G. H., and C. F. Van Loan. 2013. Matrix Computations. 4th ed. Baltimore: Johns Hopkins University Press.

Also see

[R] **regress** — Linear regression

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on citing Stata documentation.